

Хартов В.Я.
Методические указания
Для проведения лабораторных работ по курсу
Микропроцессорные системы

Лабораторная работа №3. Таймеры микроконтроллеров АТх8515

Цель работы - изучение основных режимов работы таймеров и их программирование, анализ схем включения таймеров для проведения исследований.

Микроконтроллеры AVR в зависимости от класса (Tiny, Classic, Mega) и типа модели имеют в своём составе от одного до трёх таймеров/счётчиков общего назначения T0, T1 и T2.

Первый таймер (8-разрядный T0), имеющийся во всех моделях, может использоваться для отсчёта и измерения временных интервалов или как счётчик внешних событий, а в модели ATmega8515 еще и для сравнения с заданным значением. При переполнении счётного регистра таймера генерируется запрос на прерывание. Два других таймера (16-разрядный T1 и 8-разрядный T2) имеют, кроме уже названных, дополнительные функции. Оба таймера могут генерировать запрос на прерывание не только при переполнении счётного регистра, но и при наступлении ряда других событий. Они могут также использоваться в качестве широтно-импульсных модуляторов. Кроме того, таймер T2 может работать в асинхронном (относительно тактового сигнала микроконтроллера) режиме.

Работа таймеров используемых в практикуме микроконтроллеров АТх8515 описана ниже. Каждый таймер/счётчик использует один или более выводов микроконтроллера. Эти выводы могут быть либо линиями портов ввода/вывода с альтернативными функциями, либо выделенными выводами микроконтроллера. Все выводы микроконтроллеров АТх8515, относящиеся к таймерам/счётчикам, и их функции приведены в табл.1.

Таблица 1 Выводы, используемые таймерами/счётчиками общего назначения

Название	АТх8515	STK500	Описание
T0	PB0	PB0	Вход внешнего сигнала таймера T0
T1	PB1	PB1	Вход внешнего сигнала таймера T1
ICP	ICP/PE0*	PE0	Вход захвата таймера T1
OC1A	PD5	PD5	Выход схемы сравнения таймера T1
OC1B	OC1B/PE2*	PE2	Выход схемы сравнения таймера T1

* - в микроконтроллере AT90S8515 – выделенный вывод, в ATmega8515 - вывод порта PE. При использовании линий портов ввода/вывода необходимо сконфигурировать эти выводы в соответствии с их функциональным назначением (вход или выход).

Во всех микроконтроллерах семейства AVR имеется также сторожевой таймер, который является непременным атрибутом всех современных микроконтроллеров. Этот таймер используется для предотвращения закливания программы.

1. Таймер/счётчик T0 микроконтроллера AT90S8515

Таймер/счётчик T0 (8-разрядный) может использоваться для формирования временных интервалов или для подсчёта числа внешних событий. Структурная схема таймера/счётчика T0 микроконтроллера AT90S8515 приведена на рис.1.

Таймер содержит базовый счётчик TCNT0, регистр управления TCCR0 и схему управления таймером. Кроме того, в его состав входят по одному разряду регистра запросов прерываний TIFR и маски прерываний TIMSK.

Счётчик TCNT0 доступен в любой момент времени как для чтения, так и для записи. При записи в счётчик TCNT0 во время его работы счёт будет продолжен в следующем за

командой записи машинном цикле. После подачи напряжения питания счетчик TCNT0 принимает нулевое состояние.

При переходе таймера/счётчика TCNT0 из состояния «\$FF» в состояние «\$00» устанавливается в «1» флаг TOV0 в регистре TIFR и генерируется запрос на прерывание. Разрешение прерывания осуществляется установкой в «1» разряда TOIE0 регистра маски TIMSK. Флаг общего разрешения прерывания I регистра SREG микроконтроллера также должен быть установлен в «1».




Рис.1 Структурная схема таймера/счётчика T0

Таймер/счётчик T0 может работать в двух режимах:

- 1) режим таймера. В этом режиме на вход таймера/счётчика поступают импульсы тактового сигнала микроконтроллера СК (непосредственно или через предделитель схемы управления);
- 2) режим счётчика событий. В этом режиме инкремент содержимого счётчика производится по активному фронту сигнала на входе T0 микроконтроллера (линия порта PB0).

Выбор режима работы (источника тактового сигнала), а также запуск и останов таймера/счётчика осуществляются с помощью разрядов CS02...CS00 регистра управления таймером TCCR0 (табл.2). Соответствие между состоянием этих разрядов и режимом работы таймера/счётчика приведено в табл.3. Остальные разряды регистра доступны только для чтения и содержат «0».

Таблица 2 Формат регистра TCCR0

№ разряда	7	6	5	4	3	2	1	0
Имя	-	-	-	-	-	CS02	CS01	CS00

Таблица 3. Выбор источника тактового сигнала для таймера/счётчика T0

CS02	CS01	CS00	Источник тактового сигнала
0	0	0	Таймер/счётчик остановлен
0	0	1	СК (тактовый сигнал микроконтроллера)
0	1	0	СК/8
0	1	1	СК/64
1	0	0	СК/256
1	0	1	СК/1024

1	1	0	Вывод T0, инкремент счётчика производится по спадающему фронту импульсов
1	1	1	Вывод T0, инкремент счётчика производится по нарастающему фронту импульсов

При использовании таймера/счётчика в режиме счёта внешних событий необходимо помнить, что сигнал, присутствующий на выводе T0, синхронизируется частотой тактового генератора микроконтроллера (состояние вывода T0 считывается по нарастающему фронту внутреннего тактового сигнала). Поэтому для обеспечения корректной работы таймера от внешнего сигнала промежутки времени между соседними импульсами должны быть больше периода тактового сигнала микроконтроллера.

Инкремент содержимого таймера/счётчика при работе в режиме счёта внешних событий производится даже в том случае, если вывод T0 сконфигурирован как выход. Эта особенность даёт пользователю возможность программно управлять процессом счёта.

2. Таймер/счётчик T1 микроконтроллеров ATx8515

Таймер/счётчик T1 (16-разрядный) имеет гораздо больше функций, чем таймер/счётчик T0. Прежде всего, как и таймер/счётчик T0, он может использоваться для формирования временных интервалов или для подсчёта числа внешних событий по входу T1 (линия порта PB1). Во-вторых, таймер/счётчик T1 может по внешнему сигналу сохранять своё текущее состояние в отдельном регистре ввода-вывода. В-третьих, он может выполнять определённые действия при равенстве содержимого счётного регистра и заданного значения. И, наконец, он может работать как широтно-импульсный модулятор (ШИМ). Следует иметь в виду, что генерация сигнала ШИМ «вынесена» в отдельный режим работы таймера/счётчика, в котором недоступны остальные функции (кроме генерации прерываний). В дальнейшем режим генерации сигнала ШИМ будем называть «режим ШИМ», а режим, в котором доступны остальные функции таймера/счётчика, - «режим таймера».

Структурная схема таймера/счётчика T1 приведена на рис.2.




Рис.2 Структурная схема таймера/счётчика T1

В состав таймера/счётчика входят базовый 16-разрядный счётчик TCNT1, три 16-разрядных регистра (регистр захвата ICR1 и два регистра сравнения OCR1A и OCR1B), два 16-разрядных компаратора (схемы сравнения), два 8-разрядных управляющих регистра TCCR1A и TCCR1B, а также блок управления таймером.

Все флаги состояния таймера/счётчика (переполнения, совпадения и захвата) находятся в регистре флагов прерываний от таймеров TIFR, а разрешение/запрещение прерываний от таймера осуществляется установкой/сбросом соответствующих разрядов регистра маски TIMSK.

Базовый 16-разрядный счётчик TCNT1 реализован как суммирующий (в режиме ШИМ – как суммирующий/вычитающий) счётчик и доступен в любой момент времени как для чтения, так и для записи. При записи в счётчик TCNT1 во время работы таймера счёт будет продолжен по следующему за операцией записи импульсу тактового сигнала таймера/счётчика. После подачи напряжения питания счётчик TCNT1 принимает нулевое состояние.

Физически счётчик TCNT1 размещён в двух регистрах TCNT1H:TCNT1L. Чтобы при обращении процессорного устройства микроконтроллера к этим регистрам процесс записи (чтения) обоих байтов содержимого таймера/счётчика происходил “одновременно”, обращение производится с использованием специального 8-разрядного регистра TEMP. Этот регистр используется только процессором и программно недоступен.

Собственно запись и чтение регистра TCNT1 происходят следующим образом:

- 1) при записи старшего байта значения в регистр TCNT1H он помещается в регистр TEMP. Далее, при записи младшего байта в регистр TCNT1L он объединяется с содержимым регистра TEMP, и оба байта записываются в регистр TCNT1 одновременно. Из сказанного следует, что для выполнения полного цикла записи в 16-разрядный счётчик первым должен загружаться старший байт (регистр TCNT1H).

2) при чтении регистра TCNT1L (младший байт) содержимое регистра TCNT1H пересылается в регистр TEMP. При последующем чтении регистра TCNT1H возвращается значение, сохранённое в регистре TEMP. Следовательно, для выполнения полной операции чтения 16-разрядного счетчика первым должен считываться младший байт (регистр TCNT1L).

Подобным образом с использованием регистра TEMP осуществляется обращение к остальным 16-разрядным регистрам: OCR1A, OCR1B и ICR1. Прерывания на время обращения к любому из этих регистров должны быть запрещены.

Управление таймером/счётчиком T1 осуществляется с помощью двух 8-разрядных регистров управления TCCR1A и TCCR1B. Формат регистров приведён в табл.4. Значение отдельных разрядов этих регистров будет описано далее. Неиспользуемые разряды регистров доступны только для чтения и содержат «0».

Таблица 4. Формат управляющих регистров TCCR1A, TCCR1B

Регистр\бит	7	6	5	4	3	2	1	0
TCCR1A	COM1A1	COM1A0	COM1B1	COM1B0	-	-	PWM11	PWM10
TCCR1B	ICNC1	ICES1	-	-	CTC1	CS12	CS11	CS10

По отношению к тактовому сигналу таймер/счётчик T1 может работать в двух режимах, аналогично таймеру T0. Выбор источника тактового сигнала, а также запуск и останов таймера/счётчика осуществляются с помощью разрядов CS12...CS10 регистра управления таймером TCCR1B. Соответствие между состоянием этих разрядов и режимом работы таймера/счётчика приведено в табл.5.

Таблица 5. Выбор источника тактового сигнала для таймера/счётчика T1

CS12	CS11	CS10	Источник тактового сигнала
0	0	0	Таймер/счётчик остановлен
0	0	1	СК (тактовый сигнал микроконтроллера)
0	1	0	СК/8
0	1	1	СК/64
1	0	0	СК/256
1	0	1	СК/1024
1	1	0	Вывод T1, инкремент счётчика производится по спадающему фронту импульсов
1	1	1	Вывод T1, инкремент счётчика производится по нарастающему фронту импульсов

Режим таймера

Принцип работы таймера/счётчика T1 в режиме таймера такой же, как и таймера/счётчика T0. По каждому импульсу, поступающему на тактовый вход таймера/счётчика, производится инкремент содержимого счётчика TCNT1. При переходе таймера/счётчика из состояния «\$FFFF» в состояние «\$0000» устанавливается флаг TOV1 регистра TIFR и генерируется запрос на прерывание. Разрешение прерывания осуществляется установкой в «1» разряда TOIE1 (бит 7) регистра маски TIMSK (разумеется, флаг общего разрешения прерываний I регистра SREG также должен быть установлен в «1»). Кроме

основных функций – подсчёт числа импульсов тактового сигнала микроконтроллера и подсчёт числа внешних событий, в режиме таймера T1 доступны и дополнительные функции.

Функция захвата (capture)

Данная функция заключается в сохранении в определённый момент времени состояния таймера/счётчика TCNT1 в регистре захвата ICR1. Это действие может производиться либо по активному фронту сигнала на выводе ICP микроконтроллера (линия порта PE0), либо по сигналу от аналогового компаратора, которые определяют сигнал захвата на выходе блока управления таймера (рис.2). При этом устанавливается флаг ICF1 регистра TIFR и генерируется запрос на прерывание. Разрешение прерывания осуществляется установкой в «1» разряда TICIE1 (бит 3) регистра TIMSK.

Для управления схемой захвата используются два разряда регистра TCCR1B: ICNC1 и ICES1. Разряд ICNC1 управляет схемой подавления помех. Если этот разряд сброшен в «0», схема подавления помех выключена и захват производится по первому же активному фронту на выводе ICP микроконтроллера. Если же этот разряд установлен в «1», то при появлении активного фронта на выводе ICP схема управления выполняет 4 выборки с частотой, равной тактовой частоте микроконтроллера. Захват будет выполнен только в том случае, если все выборки имеют уровень, соответствующий активному фронту сигнала (уровень «1» для нарастающего и уровень «0» для спадающего).

Активный фронт сигнала, по которому будет выполнено сохранение содержимого счётчика TCNT1 в регистре захвата, определяется состоянием разряда ICES1. Если этот разряд сброшен в «0», то активным является спадающий фронт. Если же этот разряд установлен в «1», то активным является нарастающий фронт.

Физически регистр захвата ICR1 размещён в двух регистрах ICR1H:ICR1L, доступных только для чтения. Поскольку регистр захвата является 16-разрядным, при его чтении, как уже было сказано, используется специальный регистр TEMP. При чтении регистра ICR1L (младший байт) содержимое этого регистра посылается в процессорное устройство, а содержимое регистра ICR1H (старший байт) сохраняется в регистре TEMP. При чтении регистра ICR1H возвращается значение, сохранённое в регистре TEMP. Следовательно, при чтении регистра ICR1 первым должен быть прочитан регистр ICR1L. Прерывания на время обращения к регистру ICR1 должны быть запрещены.

Функция сравнения (compare)

Данная функция заключается в непрерывном (каждый машинный цикл) сравнении содержимого счётчика TCNT1 и регистра сравнения. При совпадении их содержимого устанавливается флаг соответствующего прерывания. В микроконтроллерах ATx8515 имеется два регистра сравнения (OCR1A и OCR1B), причём операция сравнения производится независимо для каждого регистра. Если значение счётчика становится равным числу, находящемуся в регистре сравнения, то в следующем машинном цикле устанавливается соответствующий этому регистру флаг прерывания в регистре TIFR (для регистра OCR1A – флаг OCF1A, для регистра OCR1B – флаг OCF1B) и генерируется запрос на прерывание. Разрешение прерываний осуществляется установкой в «1» соответствующих разрядов регистра TIMSK (OCIE1A – бит 6 для запроса OCF1A и OCIE1B – бит 5 для запроса OCF1B).

Наряду с установкой флага в регистре TIFR при равенстве счётчика и регистра сравнения могут выполняться и другие действия: сброс таймера/счётчика (только для регистра OCR1A) и изменение состояния определённого вывода микроконтроллера (для обоих регистров).

Поведение микроконтроллера, то есть выполнение или невыполнение указанных действий, определяется несколькими разрядами регистров управления TCCR1A и TCCR1B. Состояние разрядов COM1x1, COM1x0 (x=A,B) определяет поведение вывода OC1x при совпадении содержимого счётчика TCNT1 и регистра сравнения OCR1x согласно табл.6. При изменении состояния этих разрядов соответствующее прерывание от схемы компаратора

рекомендуется запретить (во избежание ложной генерации прерывания). Чтобы таймер/счётчик мог управлять выводом, последний должен быть сконфигурирован как выход порта.

Таблица 6. Управление выводом OC1x (x=A,B)

COM1x1	COM1x0	Описание
0	0	Таймер/счётчик T1 отключен от вывода OC1x
0	1	Состояние вывода меняется на противоположное
1	0	Вывод сбрасывается в «0»
1	1	Вывод устанавливается в «1»

Если разряд CTC1 регистра управления TCCR1 установлен в «1», то при совпадении содержимого счётчика TCNT1 и регистра сравнения OCR1A производится сброс таймера/счётчика в нулевое состояние.

Каждый регистр сравнения физически размещается в двух регистрах ввода-вывода: OCR1A – OCR1AH:OCR1AL; OCR1B – OCR1BH:OCR1BL.

Поскольку регистры сравнения являются 16-разрядными, при их чтении и записи используется специальный регистр TEMP. Процесс записи и чтения 16-разрядных регистров выполняется так же, как и регистра TCNT1. Прерывания на время обращения к регистрам сравнения OCR1A и OCR1B должны быть запрещены.

Режим ШИМ (PWM)

Широтно-импульсная модуляция является одним из видов непрерывной импульсной модуляции, при которой ширина импульса пропорциональна значению модулирующего сигнала. Соответственно в данном случае широтно-импульсная модуляция заключается в генерировании сигнала с программируемой частотой и скважностью.

Для перевода таймера/счётчика T1 в режим ШИМ и задания частоты ШИМ-сигнала используются разряды PWM11:PWM10 регистра управления таймером TCCR1A. Соответствие между состоянием этих разрядов и режимом работы таймера/счётчика T1 приведено в табл.7.

Таблица 7 Управление режимом ШИМ таймера/счётчика T1

PWM11	PWM10	Описание
0	0	Режим ШИМ таймера/счётчика выключен
0	1	8-разрядный широтно-импульсный модулятор
1	0	9-разрядный широтно-импульсный модулятор
1	1	10-разрядный широтно-импульсный модулятор

Для генерации сигнала ШИМ используется схема сравнения таймера/счётчика, поэтому в микроконтроллерах ATx8515 модулятор является сдвоенным (два регистра сравнения). Названия регистров сравнения и правила обращения к ним были описаны выше. Сигнал снимается с выхода схемы сравнения таймера/счётчика.

В рассматриваемом режиме счётчик TCNT1 функционирует как реверсивный счётчик, модуль счёта которого (TOP) зависит от режима работы модулятора. Частота ШИМ-сигнала зависит от частоты тактового сигнала f_{TCK1} таймера/счётчика T1 и модуля счёта ШИМ. Значение модуля счёта и частота ШИМ-сигнала для каждого режима работы модулятора приведены в табл.8.

Таблица 8. Режимы ШИМ

Режим модулятора	Модуль счёта (TOP)	Частота ШИМ-сигнала
8-разрядный	255	$f_{TCK1}/510$

9-разрядный	511	$f_{ТСК1}/1022$
10-разрядный	1023	$f_{ТСК1}/2046$

При работе таймера/счётчика Т1 в режиме ШИМ состояние счётчика изменяется от 0 до значения TOP, а затем снова до 0, после чего цикл повторяется. При равенстве состояния счётчика и содержимого регистра сравнения состояние соответствующего этому регистру вывода микроконтроллера изменяется согласно табл.9 ("х" обозначает А или В). Таким образом, длительность ШИМ-сигнала равна $2n/f_{ТСК1}$, где n - содержимое регистра сравнения.

Таблица 9 Поведение выходов схемы сравнения в режиме ШИМ

COM1x1	COM1x0	Поведение вывода ОС1х
0	0	Таймер/счётчик Т1 отключен от вывода
0	1	Таймер/счётчик Т1 отключен от вывода
1	0	Сбрасывается в «0» при прямом счёте и устанавливается в «1» при обратном счёте (неинвертированный ШИМ-сигнал)
1	1	Устанавливается в «1» при прямом счёте и сбрасывается в «0» при обратном счёте (инвертированный ШИМ-сигнал)

Соответственно, если в регистр сравнения записать значение «0» или TOP, то при следующем совпадении состояния счётчика и содержимого регистра сравнения выход схемы сравнения переключится в устойчивое состояние согласно табл.10 (x = А или В).

Таблица 10 Устойчивые состояния выхода схемы сравнения

COM1x1	COM1x0	Регистр OCR1х	Состояние вывода ОС1х
1	0	0	0
1	0	TOP	1
1	1	0	1
1	1	TOP	0

Особенностью работы таймера/счётчика Т1 в режиме ШИМ является то, что при записи в регистр сравнения младшие 10 разрядов записываемого числа на самом деле сохраняются в специальном временном регистре. Изменение содержимого регистра сравнения происходит только в момент достижения счётчиком максимального значения TOP. Благодаря такому решению исключается появление в сигнале ШИМ импульса со случайной длительностью.

Соответственно, при чтении регистра сравнения в промежутке между записью в него и его действительным изменением возвращается содержимое временного регистра, то есть всегда возвращается значение, записанное последним.

При работе таймера/счётчика Т1 в режиме ШИМ может генерироваться прерывание по переполнению счётчика, а также прерывание от схемы сравнения. Флаги прерываний устанавливаются в «1» при изменении счётчиком направления счёта: флаг TOV1 – в точке «0», а флаги OCF1A (для регистра OCR1A) и OCF1B (для регистра OCR1B) – в точке TOP. Разрешение и обработка соответствующих прерываний выполняются как обычно.

3. Программирование таймера Т0

Режим счетчика

Подготовить программу для исследования таймера/счётчика Т0 в режиме счётчика событий. Событием в данном случае может быть замыкание одной из кнопок SWx на плате STK500. Результат работы программы отобразить средствами индикации.

С помощью 10-проводного шлейфа подключаем выводы порта PB к выводам кнопок SW0-SW7 (в данном случае будет использован только вывод PB0 – вход внешнего сигнала таймера T0). С помощью второго 10-проводного шлейфа соединяем выводы порта PD с выводами светодиодов LED0-LED7. Программируем вывод PB0 на ввод, все выводы порта PD – на вывод. Настраиваем таймер на режим счёта внешних событий (нажатие кнопки SW0). После нескольких, например четырёх, нажатий должно произойти переполнение таймера (следовательно, начальное значение счётчика TCNT0 = \$FC) и вызов обработчика прерываний. Обработчик должен включить светодиоды, показывая, что программа выполнена корректно, и заново инициализировать счётчик таймера для продолжения работы, если планируется неоднократное повторение программы. Время включения светодиодов установим, используя подпрограмму задержки.

Программа 1

```
;*****
;Программа 1 для МК ATx8515:
;демонстрация работы таймера T0 в режиме счётчика событий;
;событие - нажатие кнопки SW0.
;Соединения шлейфами: порт PB-SW, порт PD-LED
;Светодиоды включаются после четвертого нажатия на кнопку SW0
;*****
;.include "8515def.inc"           ;файл определений AT90S8515
.include "m8515def.inc"         ;файл определений ATmega8515
.def temp = r16                 ;временный регистр
;***Таблица векторов прерываний
.org $000
    rjmp INIT                   ;обработка сброса
.org $007
    rjmp T0_OVF                ;обработка переполнения таймера T0
;***Инициализация МК
INIT:    ldi temp,low(RAMEND)    ;установка
        out SPL,temp           ; указателя стека
        ldi temp,high(RAMEND)  ; на последнюю
        out SPH,temp          ; ячейку ОЗУ
        clr temp               ;инициализация выводов порта PB
        out DDRB,temp         ; на ввод
        ldi temp,0x01         ;включение 'подтягивающего' резистора
        out PORTB,temp        ; входа PB0
        ser temp               ;инициализация выводов порта PD
        out DDRD,temp         ; на вывод
        out PORTD,temp        ;выключение светодиодов
        ldi temp,0x20         ;SE=1 - разрешение перехода
        out MCUCR,temp        ; в режим Idle
;***Настройка таймера T0 на режим счётчика событий
        ldi temp,0x02         ;разрешение прерывания по
        out TIMSK,temp        ; переполнению таймера
        ldi temp,0x07         ;переключение таймера T0
        out TCCR0,temp        ; по положительному перепаду напряжения
        sei                   ;глобальное разрешение прерываний
        ldi temp,0xFC         ;$FC=-04 для
        out TCNT0,temp        ; отсчёта 4-х нажатий
LOOP:    sleep                 ;переход в режим пониженного
        nop                   ; энергопотребления
        rjmp LOOP
;***Обработка прерывания при переполнении таймера T0
T0_OVF:  clr temp
        out PORTD,temp        ;включение светодиодов
```

```

        rcall DELAY          ;задержка
        ser temp
        out PORTD,temp      ;выключение светодиодов
        ldi temp,0xFC      ;$FC=-04 для
        out TCNT0,temp     ; отсчёта 4-х нажатий
        reti
;*** Задержка ***
DELAY:  ldi r19,10
        ldi r20,255
        ldi r21,255
dd:     dec r21
        brne dd
        dec r20
        brne dd
        dec r19
        brne dd
        ret

```

Результат работы программы будет следующий. При четвёртом нажатии на кнопку SW0 загораются все светодиоды (в случае “дребезга” контактов светодиода могут включиться раньше). Длительность времени, в течение которого они горят, определяется задержкой DELAY (которая в свою очередь зависит от тактовой частоты микроконтроллера!). Далее действия могут быть повторены.

Как уже было сказано в теоретической части, события для таймера/счётчика T0 можно генерировать программно. Для этого необходимо настроить вывод PB0 как выход. В данном случае инкремент счётчика будет происходить после выполнения команд программы, эмулирующих положительный или отрицательный перепад, как этого требует настройка таймера:

```

;положительный перепад напряжения на PB0
        cbi PORTB,0
        sbi PORTB,0
;отрицательный перепад напряжения на PB0
        sbi PORTB,0
        cbi PORTB,0

```

Режим таймера

Подготовить программу для исследования таймера/счётчика T0 в режиме таймера. Для наглядности работы таймера в этом режиме можно запрограммировать следующие действия: при нажатии на первую кнопку на вход таймера поступают сигналы с частотой, равной частоте тактового генератора СК, при нажатии на вторую кнопку на вход таймера поступают сигналы с частотой, например, СК/8. В обоих случаях сразу после нажатия загораются светодиоды, а после переполнения таймера и обработки соответствующего прерывания светодиоды гаснут. Таким образом, во втором случае время свечения светодиодов будет в 8 раз больше, чем в первом, что и будет означать правильность выполнения программы. Если установить частоту тактового генератора микроконтроллера равной 512 Гц, то время включения светодиодов в первом случае будет равным 0,5 с, во втором – 4 с (0,5*8). (**Внимание!** Частоту тактового генератора необходимо изменять после программирования микроконтроллера.).

На плате STK500 необходимо соединить выводы порта PD с кнопками SW0 – SW7, выводы порта PB – со светодиодами LED0 – LED7. Для обработки нажатия кнопок используем метод последовательного опроса состояния кнопок.

Программа 2

```

;*****
;Программа 2 для МК ATx8515: демонстрация работы таймера T0
;в режиме таймера. Для наблюдений необходимо выставить частоту
;тактового генератора СК=512Гц.

```

```

;При нажатии на SW2 на вход счётчика поступают сигналы с частотой СК,
;при нажатии на SW3 - СК/8.
;В первом случае время с начала счёта до переполнения
;(выключения светодиодов) - 0,5с, во втором - 4с
;Соединения: PD2,PD3-SW2,SW3; PB-LED
;*****
;.include "8515def.inc"           ;файл определений AT90S8515
.include "m8515def.inc"         ;файл определений ATmega8515
.def temp = r16                 ;временный регистр
.equ SW2 = 2                    ;2-ой вывод порта PD
.equ SW3 = 3                    ;3-ий вывод порта PD
;***Таблица векторов прерываний
.org $000
    rjmp INIT                   ;обработка сброса
.org $007
    rjmp T0_OVF                 ;обработка переполнения таймера T0
;***Инициализация МК
INIT:    ldi temp,low(RAMEND)    ;установка
        out SPL,temp           ; указателя стека
        ldi temp,high(RAMEND)   ; на последнюю
        out SPH,temp           ; ячейку ОЗУ
        clr temp                ;инициализация 2 и 3 выводов порта PD
        out DDRD,temp           ; на ввод
        ldi temp,0x0C           ;включение 'подтягивающих' резисторов
        out PORTD,temp         ; порта PD
        ser temp                ;инициализация выводов порта PB
        out DDRB,temp           ; на вывод
        out PORTB,temp         ;выключение светодиодов
;***Настройка таймера T0 на режим таймера
        ldi temp,0x02           ;разрешение прерывания по
        out TIMSK,temp         ; переполнению таймера T0
        clr temp                ;таймер T0
        out TCCR0,temp         ; остановлен
        sei                     ;глобальное разрешение прерываний
        clr temp                ;отсчёт
        out TCNT0,temp         ; начинается с 0
;***Ожидание нажатия кнопок
WAITSET_0: sbic PIND,SW2        ;проверка нажатия
            rjmp WAITSET_1      ; кнопки SW2
;***Обработка нажатия кнопки SW2
            ldi temp,0x01       ;источник тактового сигнала - СК
            rcall LED_ON        ;включение светодиодов
WAITSET_1: sbic PIND,SW3        ;проверка нажатия
            rjmp WAITSET_0      ; кнопки SW3
;***Обработка нажатия кнопки SW3
            ldi temp,0x02       ;источник тактового сигнала - СК/8
            rcall LED_ON        ;включение светодиодов
            rjmp WAITSET_0
;***Обработка прерывания при переполнении таймера T0
T0_OVF:   ser temp
        out PORTB,temp         ;выключение светодиодов
        clr temp                ;останов
        out TCCR0,temp         ; таймера T0
        clr temp                ; таймера T0
        out TCNT0,temp         ; установка 0
        reti
;***Включение светодиодов
LED_ON:   out TCCR0,temp         ;настройка источника тактового сигнала

```

```

clr temp                ; включение
out PORTB, temp        ; светодиодов
ret

```

Результат работы программы: нажатие кнопки SW2 приводит к тому, что все светодиоды светятся в течение 0,5 с; при нажатии SW3 - в течение 4с. (Примечание. Новое нажатие будет обрабатываться сразу, то есть при нажатии сначала на SW3, а затем на SW2, не дожидаясь отключения светодиодов, они погаснут через 0,5с после второго нажатия).

Работа таймера/счётчика T1 в режимах счётчика событий и таймера аналогична работе таймера/счётчика T0. Отличие состоит в разрядности базового счётчика: TCNT0 – 8-разрядный; TCNT1 – 16-разрядный. Поэтому программы для исследования T1 в этих режимах могут быть аналогичны ранее рассмотренным.

4. Программирование функций сравнения, захвата и ШИМ таймера T1

Функция сравнения

Подготовить программу для исследования функции сравнения таймера/счётчика T1. Возможный вариант работы программы: при нажатии на кнопку START (SW0) запускается таймер – происходит инкремент счётчика с частотой СК. При совпадении значений счётчика TCNT1 и регистра сравнения OCR1B происходит изменение состояния вывода OC1B на противоположное. При совпадении значений счётчика TCNT1 и регистра сравнения OCR1A происходит изменение состояния вывода OC1A на противоположное и сброс счётного регистра в нулевое состояние. Временные диаграммы работы таймера T1 приведены на рис.3. Возможность останова таймера во время счёта реализуется с помощью внешнего прерывания от кнопки STOP (SW1). При записи значений в регистры сравнения необходимо соблюдать установленный порядок: сначала записывается старший байт, затем – младший.

Для наблюдения изменений состояний выводов OC1A и OC1B их необходимо соединить с выводами светодиодов. Частота тактового генератора устанавливается равной 256 Гц. Коммутация осуществляется 2-проводными шнурами: LED0 – PE2, LED1 – PD5, SW0 – PD0, SW1 – PD2.




Рис.3. Временные диаграммы работы таймера/счётчика T1 при исследовании функции сравнения

Программа 3

```

; *****
; Программа 3 для МК ATx8515:
; демонстрация работы функции сравнения таймера T1
; Для наглядности необходимо выставить частоту тактового
; генератора СК=256 Гц.
; При нажатии на SW0 (START) происходит инкремент счётчика с частотой СК,
; при нажатии на SW1 (STOP) счётчик останавливается.
; При совпадении содержимого счётчика и регистра сравнения
; OCR1B происходит переключение светодиода LED0,

```

```

;счётного регистра и регистра сравнения OCR1A - LED1.
;Соединения: LED0-PE2, LED1-PD5, SW0-PD0, SW1-PD2
;*****
;.include "8515def.inc" ;файл определений AT90S8515
.include "m8515def.inc" ;файл определений ATmega8515
.def temp = r16 ;временный регистр
.equ START = 0 ;0-ой вывод порта PD
.org $000
    rjmp INIT ;обработка сброса
.org $003
    rjmp STOP_PRESSED ;обработка внешнего прерывания INT0 -
; ; нажатие STOP
;***Инициализация МК
INIT:    ldi temp,low(RAMEND) ;установка
        out SPL,temp ; указателя стека
        ldi temp,high(RAMEND) ; на последнюю
        out SPH,temp ; ячейку ОЗУ
        ldi temp,0x20 ;инициализация выводов порта PD:
        out DDRD,temp ; 0,2 - на ввод, 5 - на вывод
        ldi temp,0x05 ;включение 'подтягивающих'
        out PORTD,temp ; резисторов порта PD и выключение СД
        ldi temp,0x04 ;/// для ATmega8515 инициализация вывода порта
        out DDRE,temp ;/// PE2 (OC1B) на вывод!
        ldi temp,0x40 ;разрешение прерывания INT0
        out GIMSK,temp ; в регистре GIMSK (или GICR)
        clr temp ;обработка прерывания INT0
        out MCUCR,temp ; по низкому уровню
;***Настройка функции сравнения таймера T1
        ldi temp,0x00 ;запрещение прерываний
        out TIMSK,temp ; от таймера
        cli ;глобальное запрещение прерываний
        ldi temp,0x50 ;при равенстве состояния выводов OC1A и
        out TCCR1A,temp ; OC1B изменяются на противоположные
        ldi temp,0x00 ;таймер
        out TCCR1B,temp ; остановлен
        ldi temp,0x00 ;запись числа в
        out OCR1BH,temp ; регистр сравнения,
        ldi temp,0x80 ; первым записывается
        out OCR1BL,temp ; старший байт
        ldi temp,0x00 ;запись числа в
        out OCR1AH,temp ; регистр сравнения,
        ldi temp,0xFF ; первым записывается
        out OCR1AL,temp ; старший байт
        clr temp ;обнуление
        out TCNT1H,temp ; содержимого
        out TCNT1L,temp ; счётного регистра
        sei ;глобальное разрешение прерываний
WAITSTART: sbic PIND,START ;ожидание нажатия
            rjmp WAITSTART ; кнопки START
            ldi temp,0x09 ;запуск таймера, при
            out TCCR1B,temp ; совпадении с OCR1A - сброс
LOOP:      nop ;во время цикла происходит
            rjmp LOOP ; увеличение содержимого счётного регистра
;***Обработка прерывания от кнопки STOP
STOP_PRESSED:
        ldi temp,0x08 ;остановка
        out TCCR1B,temp ; таймера
WAITSTART_2: ;ожидание

```

```

sbic PIND,START      ; нажатия
rjmp WAITSTART_2    ; кнопки START
ldi temp,0x09       ; запуск
out TCCR1B,temp     ; таймера
reti

```

Результат работы программы соответствует диаграммам на рис.3. При нажатии на кнопку SW0 светодиоды работают в следующей последовательности: оба светодиода погашены, далее включается LED0, затем LED1, выключается LED0, затем LED1 и т.д. В любой момент процесс можно остановить нажатием кнопки SW1.

Функция захвата

Подготовить программу для исследования функции захвата таймера/счётчика T1. Захват должен происходить при нажатии соответствующей кнопки. По событию «захват» должен вызываться обработчик прерывания, который переписывает содержимое 16-разрядного регистра захвата ICR1 в регистры для хранения младшего и старшего байта. Пересылка данных в порт для индикации возможна только из регистров общего назначения, поэтому должны использоваться регистры хранения байтов. Вывод каждого байта может быть связан с нажатием отдельной кнопки. С помощью регистра сравнения OCR1A можно задать максимальное значение, которое можно получить в счётчике и, следовательно, в регистре захвата. Для этого необходимо настроить работу таймера так, чтобы он сбрасывался при равенстве значений счётчика TCNT1 и регистра сравнения, установив бит CTC1 регистра TCCR1B в «1». Это будет дополнительным признаком правильности работы программы. Так например, если содержимое OCR1A равно \$00FF, а бит CTC1 установлен в «1», то в старшем байте регистра захвата всегда будет 0.

При моделировании программы в AVR Studio необходимо помнить, что вывод ICP в микроконтроллере AT90S8515 является выделенным, а в микроконтроллере ATmega8515 – это линия порта ввода/вывода PE0. Вызов обработчика прерывания по событию «захват» можно осуществить, установив в «1» флаг TICIE1 регистра маски TIMSK.

Коммутация кнопок с выводами микроконтроллера осуществляется 2-проводными шнурами: SW0 – PD0, SW1 – PD1, SW2 - PD2, SW3 – PE0. Светодиоды подключаются к выводам порта PB 10-проводным шлейфом.

Программа 4

```

; *****
; Программа 4 для МК ATx8515:
; демонстрация работы функции захвата таймера T1
; Для наблюдения необходимо выставить частоту тактового
; генератора СК=256 Гц.
; При нажатии на SW0 (START) на вход счётчика поступает сигнал с частотой
; СК, при нажатии на SW3 (CAPT) происходит захват состояния таймера.
; При совпадении содержимого счётчика и регистра сравнения OCR1A
; происходит сброс таймера.
; При нажатии на SW1 (SHOW_L) на светодиоды выводится значение младшего
; байта регистра захвата, SW2 (SHOW_H) – старшего байта регистра захвата.
; Соединения: SW0-PD0, SW1-PD1, SW2-PD2, SW3-PE0, 10-проводным шлейфом PB-LED
; *****
.include "8515def.inc"      ; файл определений AT90S8515
.include "m8515def.inc"    ; файл определений ATmega8515
.def temp = r16            ; временный регистр
.def H_byte = r17          ; для хранения старшего байта
.def L_byte = r18          ; для хранения младшего байта
.equ START = 0             ; 0-ой вывод порта PD
.equ SHOW_L = 1            ; 1-ый вывод порта PD
.equ SHOW_H = 2            ; 2-ой вывод порта PD

```

```

;***Векторы прерываний
.org $000
    rjmp INIT                ;обработка сброса
.org $003
    rjmp CAPT_PRESSED       ;обработка внешнего прерывания INT0 -
                            ;нажатие CAPT

;***Инициализация МК
INIT:    ldi temp,low(RAMEND) ;установка
        out SPL,temp        ; указателя стека
        ldi temp,high(RAMEND) ; на последнюю
        out SPH,temp        ; ячейку ОЗУ
        ldi temp,0x00       ;инициализация выводов
        out DDRD,temp       ; порта PD на ввод
        ldi temp,0x07       ;включение 'подтягивающих'
        out PORTD,temp      ; резисторов порта PD
        ;ldi temp,0x00      ;/// для ATmega8515 инициализация вывода порта
        ;out DDRE,temp      ;/// PE0 (ICP) на ввод!
        ser temp            ;инициализация выводов
        out DDRB,temp       ; порта PB на вывод
        out PORTB,temp      ;выключение светодиодов
        cli                 ;глобальное запрещение прерываний
        ldi temp,0x00       ;отключение от таймера
        out TCCR1A,temp     ; выводов
        ldi temp,0x00       ;таймер
        out TCCR1B,temp     ; остановлен
        ldi temp,0xFF       ;запись числа в
        out OCR1AH,temp     ; регистр сравнения,
        ldi temp,0xFF       ; первым записывается
        out OCR1AL,temp     ; старший байт
        clr temp           ;обнуление
        out TCNT1H,temp     ; содержимого
        out TCNT1L,temp     ; счётного регистра
        ldi L_byte,0xF0     ;начальные значения
        ldi H_byte,0x0F     ; для проверки
        sei                 ;глобальное разрешение прерываний
WAITSTART: sbic PIND,START ;ожидание нажатия
            rjmp WAITSTART ; кнопки START
            ldi temp,0x08   ;разрешение прерывания
            out TIMSK,temp  ; по событию "захват" таймера
            ldi temp,0xC9   ;запуск таймера, при
            out TCCR1B,temp ; совпадении с OCR1A - сброс
WAIT_L:    sbic PIND,SHOW_L ;ожидание нажатия
            rjmp WAIT_H     ; кнопки SHOW_L - показать младший байт
            out PORTB,L_byte ;вывод на СД
WAIT_H:    sbic PIND,SHOW_H ;ожидание нажатия
            rjmp WAIT_L     ; кнопки SHOW_H - показать старший байт
            out PORTB,H_byte ;вывод на СД
            rjmp WAIT_L

;***Обработка прерывания от кнопки CAPT
CAPT_PRESSED:
        in L_byte,ICR1L     ;считывание младшего байта
        in H_byte,ICR1H     ;считывание старшего байта
        com L_byte         ;инвертирование для вывода
        com H_byte         ;инвертирование для вывода
        reti

```

Режим ШИМ

Подготовить программу для исследования работы таймера T1 в режиме ШИМ. Выводы OC1A (PD5) и OC1B (PE2) необходимо подключить к светодиодам (PD5-LED0, PE2-LED1), выводы порта PD0-PD3 к кнопкам общего назначения SW0-SW3 соответственно. Варианты обработки нажатия кнопок:

- 1) при нажатии на SW0 на выходах OC1A и OC1B устанавливаются значения «0» и «1» соответственно;
- 2) при нажатии на SW1 происходит генерация ШИМ-сигнала со скважностью F1 (скважность зависит от значения в регистре сравнения);
- 3) при нажатии на SW2 происходит генерация ШИМ-сигнала со скважностью F2;
- 4) при нажатии на SW3 на выходах OC1A и OC1B устанавливаются «1» и «0» соответственно.

Временная диаграмма работы ШИМ приведена на рис.4.




Рис.4 Формирование ШИМ-сигнала

Программа 5

```
;*****  
;Программа 5 для МК ATx8515:  
;демонстрация работы таймера T1 в режиме ШИМ  
;Для наглядности необходимо выставить частоту тактового  
;генератора = 2048 Гц.  
;При нажатии на SW0 (SHOW_0) на выходах OC1A и OC1B устанавливаются 0 и 1  
;При нажатии на SW1 (SHOW_F1) происходит генерация  
;ШИМ-сигнала со скважностью F1  
;При нажатии на SW2 (SHOW_F2) происходит генерация  
;ШИМ-сигнала со скважностью F2  
;При нажатии на SW3 (SHOW_1) на выходах OC1A и OC1B устанавливаются 1 и 0  
;Соединения: PD5-LED0, PE2-LED1, PD0-SW0, PD1-SW1, PD2-SW2, PD3-SW3  
;*****  
;.include "8515def.inc" ;файл определений AT90S8515  
.include "m8515def.inc" ;файл определений ATmega8515  
.def temp = r16 ;временный регистр  
;***Выводы порта PD  
.equ SHOW_0 = 0  
.equ SHOW_F1 = 1  
.equ SHOW_F2 = 2  
.equ SHOW_3 = 3  
.org $000  
  
 rjmp INIT ;обработка сброса  
;***Инициализация МК  
INIT: ldi temp,low(RAMEND) ;установка  
 out SPL,temp ; указателя стека  
 ldi temp,high(RAMEND) ; на последнюю
```

```

out SPH,temp          ; ячейку ОЗУ
ldi temp,0x20         ;инициализация выводов порта PD:
out DDRD,temp         ; 0-3 - на ввод, 5 - на вывод
ldi temp,0x0F         ;включение 'подтягивающих'
out PORTD,temp        ; резисторов порта PD
ldi temp,0x04        ;/// для ATmega8515 инициализация вывода порта
out DDRE,temp        ;/// PE2 (OC1B) на вывод!
cli                   ;глобальное запрещение прерываний
ldi temp,0xB3         ;настройка таймера на ШИМ и работу с
out TCCR1A,temp       ; выводами OC1A и OC1B
clr temp              ;обнуление
out OCR1AH,temp       ; регистров
out OCR1AL,temp       ; сравнения и
out OCR1BH,temp       ; ...
out OCR1BL,temp       ; ...
out TCNT1H,temp       ; счётного
out TCNT1L,temp       ; регистра
ldi temp,0x01         ;таймер
out TCCR1B,temp       ; запущен: частота = СК
sei                   ;глобальное разрешение прерываний

WAIT_0:
sbic PIND,SHOW_0     ; нажатия
rjmp WAIT_F1         ; кнопки SHOW_0,устойчивые состояния 0 и 1
;***Перевод в устойчивые состояния выводов OC1A=0, OC1B=1
clr temp              ;запись числа в
out OCR1AH,temp       ; регистр сравнения, первым
out OCR1AL,temp       ; записывается старший байт
out OCR1BH,temp       ;запись числа в регистр сравнения, первым
out OCR1BL,temp       ; записывается старший байт

WAIT_F1:
sbic PIND,SHOW_F1    ; ожидание нажатия
rjmp WAIT_F2         ; кнопки SHOW_F1 -режим ШИМ скважностью F1
;***Настройка таймера на режим ШИМ со скважностью F1
ldi temp,0x00         ;запись числа в
out OCR1AH,temp       ; регистры сравнения,
out OCR1BH,temp       ; первым записывается
ldi temp,0xFF         ; старший байт
out OCR1AL,temp
out OCR1BL,temp

WAIT_F2:
sbic PIND,SHOW_F2    ; ожидание нажатия
rjmp WAIT_3          ; кнопки SHOW_F2 -режим ШИМ скважностью F2
;***Настройка таймера на режим ШИМ со скважностью F2
ldi temp,0x02         ;запись числа в
out OCR1AH,temp       ; регистры сравнения,
out OCR1BH,temp       ; первым записывается
ldi temp,0xFF         ; старший байт
out OCR1AL,temp
out OCR1BL,temp

WAIT_3:
sbic PIND,SHOW_3     ; ожидание нажатия
rjmp WAIT_0          ; кнопки SHOW_3,устойчивые состояния 1 и 0
;***Перевод в устойчивые состояния выводов OC1A=1, OC1B=0
ser temp              ;запись числа в
out OCR1AH,temp       ; регистр сравнения, первым
out OCR1AL,temp       ; записывается старший байт
out OCR1BH,temp       ;запись числа в регистр сравнения, первым
out OCR1BL,temp       ; записывается старший байт
rjmp WAIT_0

```

Результат соответствует диаграммам на рис.4.4. При нажатии на SW0 загорается только LED0, при нажатии на SW3 – только LED1. При нажатии на SW1 или SW2 светодиоды попеременно включаются/выключаются со скважностью F1 или F2 соответственно.

5. Сторожевой таймер

Основная функция сторожевого таймера (*Watchdog Timer*) – защита устройства от сбоев. Благодаря сторожевому таймеру можно прервать выполнение зациклившейся программы или выйти из других непредвиденных ситуаций, препятствующих нормальному выполнению программы. Структурная схема сторожевого таймера приведена на рис.5.




Рис.5 Структурная схема сторожевого таймера

Для управления сторожевым таймером предназначен регистр WDTCR. Формат регистра приведён в табл.11.

Таблица 11 Формат регистра WDTCR

№ разряда	7	6	5	4	3	2	1	0
Имя	-	-	-	WDTOE	WDE	WDP2	WDP1	WDP0

Для включения/выключения сторожевого таймера используются два разряда регистра WDTCR – WDE и WDTOE. Если разряд WDE установлен в «1», сторожевой таймер включен, если сброшен в «0» - выключен. Непосредственно перед включением таймера рекомендуется также выполнить его сброс командой WDR.

Чтобы избежать непреднамеренного выключения таймера, предназначен разряд WDTOE. Дело в том, что выключение сторожевого таймера (сброс разряда WDE) можно осуществить только при установленном разряде WDTOE, который через 4 машинных цикла после установки в «1» аппаратно сбрасывается. Благодаря этому возможность случайного выключения сторожевого таймера практически исключается.

Исходя из сказанного, для выключения сторожевого таймера рекомендуется следующая последовательность действий:

- 1) одной командой записать «1» в разряды WDE и WDTOE,
- 2) в течении следующих четырёх машинных циклов записать «0» в разряд WDE.

Период наступления тайм-аута сторожевого таймера задаётся с помощью разрядов WDP2-WDP0 регистра WDTCR согласно табл.12.

Таблица 12. Задание периода сторожевого таймера для AT90S8515

WDP2	WDP1	WDP0	Число тактов генератора	Период наступления тайм-аута (типичное значение)	
				$V_{cc} = 3.0 \text{ В}$	$V_{cc} = 5.0 \text{ В}$

0	0	0	16·1024	47 мс	15 мс
0	0	1	32·1024	91 мс	30 мс
0	1	0	64·1024	0.19 с	60 мс
0	1	1	128·1024	0.38 с	0.12 с
1	0	0	256·1024	0.75 с	0.24 с
1	0	1	512·1024	1.5 с	0.49 с
1	1	0	1024·1024	3.0 с	0.97 с
1	1	1	2048·1024	6.0 с	1.9 с

Чтобы избежать непреднамеренного сброса микроконтроллера при изменении периода сторожевого таймера, необходимо перед записью разрядов WDP2-WDP0 либо запретить работу сторожевого таймера, либо сбросить его.

Программирование сторожевого таймера

Подготовить программу для исследования сторожевого таймера. Полагая, что при нажатии на кнопку SW0 тайм-аут таймера наступает через 0.49 с, а при нажатии на SW1 – через 1.9 с, выберем из табл.4.11 ($V_{CC}=5.0$ В) значения загружаемых коэффициентов согласно заданному периоду сторожевого таймера. В обоих случаях до наступления тайм-аута светодиоды должны быть включены, после – происходит сброс микроконтроллера, и светодиоды гаснут.

Программа для микроконтроллера приведена ниже. (**Внимание!** Перед программированием микроконтроллера ATmega8515 в окне STK500 AVR Studio 4 на вкладке Fuses установить флажок S8515C и запрограммировать конфигурационную ячейку.).

Коммутация выводов: SW0 – PD0, SW1 – PD1, LED – PB 10-проводным шлейфом.

Программа 6

```

;*****
;Программа 6 для МК AT9x8515:
;демонстрация работы сторожевого таймера
;Частота тактового генератора не существенна, так как сторожевой
;таймер имеет независимый генератор
;При нажатии на SW0 (PERIOD_1) наступление тайм-аута происходит через 0,49с
;(время включения светодиодов)
;При нажатии на SW1 (PERIOD_2) наступление тайм-аута происходит через 1,9с
;(время включения светодиодов)
;Соединения:PD0-SW0,PD1-SW1, PB-LED(10-проводной шлейф)
;*****
;.include "8515def.inc"           ;файл определений AT90S8515
.include "m8515def.inc"         ;файл определений ATmega8515
.def temp = r16                 ;временный регистр
;***Выводы порта PD
.equ PERIOD_1 = 0
.equ PERIOD_2 = 1
.org $000

        rjmp INIT               ;обработка сброса
;***Инициализация МК
INIT:    ldi temp,low(RAMEND)    ;установка
        out SPL,temp           ; указателя стека
        ldi temp,high(RAMEND)   ; на последнюю
        out SPH,temp           ; ячейку ОЗУ
        clr temp                ;инициализация выводов порта PD
        out DDRD,temp          ; на ввод
        ldi temp,0x03           ;включение `подтягивающих'
        out PORTD,temp         ; резисторов порта PD

```

```

ser temp                ;инициализация выводов порта PB
out DDRB,temp           ; на вывод
out PORTB,temp         ;выключение светодиодов
ldi temp,$18           ;выключение
out WDTCR,temp         ; сторожевого
ldi temp,$10           ; таймера:
out WDTCR,temp         ; WDE=0
WAIT_SW0: sbic PIND,PERIOD_1 ;ожидание нажатия
rjmp WAIT_SW1         ; кнопки PERIOD_1
;***Назначение периода наступления тайм-аута = 0,49с
clr temp               ;включение
out PORTB,temp         ; светодиодов
ldi temp,$0D           ;включение сторожевого таймера,
out WDTCR,temp         ; период 0,49с
WAIT_SW1: sbic PIND,PERIOD_2 ;ожидание нажатия
rjmp WAIT_SW0         ; кнопки PERIOD_2
;***Назначение периода наступления тайм-аута = 1,9с
clr temp               ;включение
out PORTB,temp         ; светодиодов
ldi temp,$0F           ;включение сторожевого таймера,
out WDTCR,temp         ; период 1,9с
rjmp WAIT_SW0

```

Задания для самостоятельного программирования

Задание 1. Изменить программу 1, добавив в нее обработку нажатия кнопки, исключаящую влияние “дребезга”.

Для этого запрограммировать линию порта PB0 на вывод для программного ввода событий в таймер. Ввод событий выполнять при условии замыкания кнопки SW_x, присоединенной к порту (линию и порт выбрать самостоятельно). Выполняя проверку состояния кнопки SW_x, эмулировать сигнал «1» на выводе PB0 при замкнутом состоянии «0» кнопки. Перед загрузкой программы выполнить ее отладку в AVR Studio. Убедившись в правильности работы программы, проверить ее в STK500.

Задание 2. Подготовить программу по примеру 4.2 для проверки работы таймера/счётчика T1 в режиме таймера. Время включения светодиодов 1с и 8с.

Контрольные вопросы

1. Какова скважность сигналов, формируемых программой 4.3? Как изменяется временная диаграмма выходного сигнала при изменении значения, загружаемого в регистр сравнения?
2. В каком порядке выполняется запись байтов в 16 - разрядные регистры?
3. В каком порядке выполняется чтение байтов из 16 - разрядного счетчика TCNT1, регистра захвата ICR1?
4. Какие значения необходимо занести в регистры сравнения в программе 5 для формирования ШИМ - сигналов скважностью 4 (отношения периода к длительности сигнала) при том же периоде ШИМ - сигналов?
5. Как изменится скважность ШИМ - сигналов, формируемых при работе программы 4.5, если изменить управляющее слово в регистре TCCR1A на B1 (B2)?
6. Назовите максимальное время наступления тайм-аута у сторожевого таймера ATx8515.
7. Назовите максимальное время захвата, которое можно зарегистрировать при работе программы 4. В каких приложениях можно применить эту программу?